Pensieve header: Verifying g1 identities for McGill-1702.

Reminders from GWU-1612.

1-Smidgen sl_2 Let $\mathfrak{g}_1$ be the 4-dimensional Lie algebra $\mathfrak{g}_1 = \langle h, e, l, f\rangle$ over the ring $R = \mathbb{Q}[\epsilon]/(\epsilon^2 = 0)$, with h central and with $[f,l] = f$, $[e,l] = -e$, and $[e,f] = h - 2\epsilon l$.

Implementing g_1

```
ϵ /: ϵ^k_ /; k > 1 := 0;
PBWRule = {e → 1, l → 2, f → 3};
B[U@e, U@l] = -U@e;
B[U@f, U@l] = U@f;
B[U@e, U@f] = h U[] - 2 ϵ U@l;
```

```
$TD = 3;
ħ /: ħ^d_. /; d > $TD := 0;
```

```
x_ ≤ y_ := OrderedQ[{x, y} /. PBWRule];
x_ < y_ := ! OrderedQ[{y, x} /. PBWRule];
Simp[ℰ_] := Collect[ℰ, _U, Expand];
```

```
U_i_[ℰ_] := ℰ /. {h → h_i, t → t_i, u_U ⧴ Replace[u, x_ ⧴ x_i, 1]};
B[U[(x_)_i_], U[(y_)_i_]] := B[U[x_i], U[y_i]] = U_i[B[U@x, U@y]];
B[U[(x_)_i_], U[(y_)_j_]] /; i =!= j := 0;
B[x_, x_] = 0;
B[U[y_], U[x_]] := B[U[y], U[x]] = Simp[-B[U[x], U[y]]];
B[x_, y_] := x ** y - y ** x;
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

```
U[xx___, x_] ** U[y_, yy___] := If[x ≤ y, U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
```

```
UU[l___, x_^n_, r___] := UU[l, Sequence @@ Table[x, {n}], r];
UU[l___, 1, r___] := UU[l, r];
UU[] = U[];
UU[l_, r___] := U[l] ** UU[r];
```

```
UProducts[{}, 0] = {UU[]};
UProducts[{}, n_Integer] /; n > 0 = {};
UProducts[{x_, xs___}, n_Integer] :=
  Sort@Flatten@Table[UU[x^k] ** u, {k, 0, n}, {u, UProducts[{xs}, n - k]}];
UProducts[xs_List, k_Integer, n_Integer] := UProducts[Flatten@Table[x_j, {x, xs}, {j, k}], n];
UProducts[any__, {n_}] := Flatten@Table[UProducts[any, k], {k, 0, n}];
```

```
r_{i_,j_} := Simp[ℏ (h_i UU[l_j] - ϵ UU[l_i, l_j] + UU[e_i, f_j])]
```

```
UExp[u_] := Module[{s, t, k},
   s = t = U[]; k = 0;
   While[k < 20 ∧ 0 =!= (t = t ** u), s += t / (++k) !];
   Simp[s]
  ];
R_{i_,j_} := UExp[r_{i,j}];
```

```
𝕆[poly_, specs___] := Module[{vs, us, z},
  vs = Join @@ (First /@ {specs});
  us = Join @@ ({specs} /. (l_ → s_) ⧴ (l /. x_i_ ⧴ x_s));
  Simp@Total[CoefficientRules[Normal@Series[poly, {ℏ, 0, $TD}], vs] /. (p_ → c_) ⧴ c UU @@ (us^p)]
 ]
```

Testing YBE

```
UExp[ℏ U@e_1]
```

$$U[\,] + \hbar\, U[e_1] + \frac{1}{2}\hbar^2 U[e_1, e_1] + \frac{1}{6}\hbar^3 U[e_1, e_1, e_1]$$

```
R_{1,2}
```

$$U[\,] + \hbar h_1 U[l_2] + \left(\hbar + \frac{\hbar^2 h_1}{2} + \frac{1}{6}\hbar^3 h_1^2\right) U[e_1, f_2] - \epsilon\hbar\, U[l_1, l_2] + \frac{1}{2}\hbar^2 h_1^2 U[l_2, l_2] + \left(-\frac{\epsilon\hbar^2}{2} - \frac{1}{3}\epsilon\hbar^3 h_1\right) U[e_1, l_1, f_2] + \left(-\frac{\epsilon\hbar^2}{2} + \hbar^2 h_1 - \frac{1}{6}\epsilon\hbar^3 h_1 + \frac{1}{2}\hbar^3 h_1^2\right) U[e_1, l_2, f_2] - \epsilon\hbar^2 h_1 U[l_1, l_2, l_2] + \frac{1}{6}\hbar^3 h_1^3 U[l_2, l_2, l_2] + \left(\frac{\hbar^2}{2} - \frac{\epsilon\hbar^3}{6} + \frac{\hbar^3 h_1}{2}\right) U[e_1, e_1, f_2, f_2] + \left(-\epsilon\hbar^2 - \epsilon\hbar^3 h_1\right) U[e_1, l_1, l_2, f_2] + \left(-\frac{1}{2}\epsilon\hbar^3 h_1 + \frac{1}{2}\hbar^3 h_1^2\right) U[e_1, l_2, l_2, f_2] - \frac{1}{2}\epsilon\hbar^3 h_1^2 U[l_1, l_2, l_2, l_2] - \frac{1}{2}\epsilon\hbar^3 U[e_1, e_1, l_1, f_2, f_2] + \left(-\frac{\epsilon\hbar^3}{2} + \frac{\hbar^3 h_1}{2}\right) U[e_1, e_1, l_2, f_2, f_2] - \epsilon\hbar^3 h_1 U[e_1, l_1, l_2, l_2, f_2] + \frac{1}{6}\hbar^3 U[e_1, e_1, e_1, f_2, f_2, f_2] - \frac{1}{2}\epsilon\hbar^3 U[e_1, e_1, l_1, l_2, f_2, f_2]$$

```
$TD = 2; Simp[R_{1,2} ** R_{1,3} ** R_{2,3} - R_{2,3} ** R_{1,3} ** R_{1,2}]
```

0

```
$TD = 3; Simp[R_{1,2} ** R_{1,3} ** R_{2,3} - R_{2,3} ** R_{1,3} ** R_{1,2}]
```

0

```
$TD = 4; Simp[R_{1,2} ** R_{1,3} ** R_{2,3} - R_{2,3} ** R_{1,3} ** R_{1,2}]
```

g_0 tests

```
$TD = 6; 𝕆[Exp[ℏ h_1 l_2 + (e^(ℏ h_1) - 1)/h_1 e_1 f_2], {e_1} → 1, {l_2, f_2} → 2] == R_{1,2} /. ϵ → 0
```

True

```
$TD = 3; 𝕆[e^(ℏ δ e_1 f_1), {f_1, e_1} → 1] /. ϵ → 0
```

$$\left(1 - \delta\hbar h_1 + \delta^2\hbar^2 h_1^2 - \delta^3\hbar^3 h_1^3\right) U[\,] + \left(\delta\hbar - 2\delta^2\hbar^2 h_1 + 3\delta^3\hbar^3 h_1^2\right) U[e_1, f_1] + \left(\frac{\delta^2\hbar^2}{2} - \frac{3}{2}\delta^3\hbar^3 h_1\right) U[e_1, e_1, f_1, f_1] + \frac{1}{6}\delta^3\hbar^3 U[e_1, e_1, e_1, f_1, f_1, f_1]$$

$TD = 3; With[{$\nu$ = $(1 + \hbar\, h_1\, \delta)^{-1}$}, $\mathbb{O}[\nu\, e^{\hbar\,\nu\,\delta\, e_1 f_1}$, {$e_1$, f_1} → 1]] /. ϵ → 0

$$\left(1 - \delta\,\hbar\, h_1 + \delta^2\,\hbar^2\, h_1^2 - \delta^3\,\hbar^3\, h_1^3\right) U[\,] + \left(\delta\,\hbar - 2\,\delta^2\,\hbar^2\, h_1 + 3\,\delta^3\,\hbar^3\, h_1^2\right) U[e_1, f_1] + \left(\frac{\delta^2\,\hbar^2}{2} - \frac{3}{2}\,\delta^3\,\hbar^3\, h_1\right) U[e_1, e_1, f_1, f_1] + \frac{1}{6}\,\delta^3\,\hbar^3\, U[e_1, e_1, e_1, f_1, f_1, f_1]$$

$TD = 6; With[{$\nu$ = $(1 + \hbar\, h_1\, \delta)^{-1}$}, $\mathbb{O}[e^{\hbar\,\delta\, e_1 f_1}$, {$f_1$, e_1} → 1] == $\mathbb{O}[\nu\, e^{\hbar\,\nu\,\delta\, e_1 f_1}$, {$e_1$, f_1} → 1] /. ϵ → 0]

True

g_1 tests

Λ[ν_] := Expand[

$$\frac{1}{2\,\mu^4}\left(-b\,\alpha^2\,\beta^2 + u^2\,\beta^2\,\delta\,(2 + b\,\delta) + w^2\,\delta\,(\alpha + u\,\delta)\,(\alpha\,(2 + b\,\delta) + u\,\delta\,(4 + 3\,b\,\delta)) - 4\,b\,\alpha\,\beta\,\delta\,\mu + 4\,c\,\alpha\,\beta\,\mu^2 - 2\,b\,\delta^2\,\mu^2 + 4\,c\,\delta\,\mu^3 + 2\,u\,\beta\,(\alpha\,\beta + 2\,\delta\,\mu\,(1 + c\,\mu)) + 2\,w\,(\alpha^2\,\beta + 2\,\alpha\,\delta\,(u\,\beta\,(2 + b\,\delta) + \mu\,(1 + c\,\mu)) + u\,\delta^2\,(u\,\beta\,(3 + 2\,b\,\delta) + 2\,\mu\,(2 + b\,\delta + c\,\mu)))\right)$$ /.

{α | β → 0, μ → ν^{-1}, b → $\hbar\, h_1$, c → l_1, u → $\hbar\, e_1$, w → f_1}

]

$TD = 6; With[

{ν = $(1 + \hbar\, h_1\, \delta)^{-1}$},

Simp[$\mathbb{O}[e^{\hbar\,\delta\, e_1 f_1}$, {$f_1$, e_1} → 1] − $\mathbb{O}[\nu\,(1 + \epsilon\,\hbar\,\Lambda[\nu])\, e^{\hbar\,\nu\,\delta\, e_1 f_1}$, {$e_1$, l_1, f_1} → 1]]

]

0

Λ[ν] /. {ℏ → 1, $x_{\_1}$ ⧴ x}

$$2\, l\,\delta\,\nu - h\,\delta^2\,\nu^2 + 2\, e\, f\, l\,\delta^2\,\nu^2 + 4\, e\, f\,\delta^2\,\nu^3 + 2\, e\, f\, h\,\delta^3\,\nu^3 + 2\, e^2\, f^2\,\delta^3\,\nu^4 + \frac{3}{2}\, e^2\, f^2\, h\,\delta^4\,\nu^4$$

TeXForm[Λ[ν] /. {ℏ → 1, $x_{\_1}$ ⧴ x}]

```
2 \delta ^3 e^2 f^2 \nu ^4+\frac{3}{2} \delta ^4 e^2 f^2 h \nu ^4+4 \delta ^2 e f \nu ^3+2 \delta ^3 e f h
   ^2 h \nu ^2+2 \delta  l \nu
```